

Решения уравнения КдФ в виде бегущей волны

В.Э. Адлер. Классические интегрируемые системы, весенний семестр 2020

Проанализируем простейшее семейство точных решений уравнения КдФ

$$u_t = u_3 + 6 u u_1,$$

а именно, решения вида $u = -2 y(X)$, $X = x - c t$. Множитель -2 введён здесь для удобства. Функция $y(X)$ удовлетворяет ОДУ порядка 3, которое легко два раза проинтегрировать:

$$y''' - 12 y y' + c y' = 0 \Rightarrow$$

$$y'' - 6 y^2 + c y + c_1 = 0 \Rightarrow \text{(интегрирующий множитель } 2 y')$$

$$(y')^2 = 4 R(y) = 4 y^3 - c y^2 - 2 c_1 y - c_2.$$

Решение последнего уравнения можно записать в неявном виде, как

$$\int \frac{dy}{\sqrt{R(y)}} = 2 X + d,$$

то есть, задача сводится к обращению интеграла в левой части. Вообще говоря, это приводит к эллиптическим функциям, но если корни кратные, то ответ упрощается.

Отметим следующие точечные преобразования, действующие на решениях КдФ. Благодаря им, корни R можно сдвигать и масштабировать, так что остаётся не более одного существенного параметра.

Проверьте, что каждое из этих преобразований образует однопараметрическую группу, действующую на переменных x, t, u (во втором случае нужно дополнительно переобозначить $a \rightarrow \exp(a)$).

Утверждение. Если $u(x, t)$ удовлетворяет уравнению КдФ, то это верно также для функций

$$U(x, t) = u(x + 6 a t, t) + a \text{ (преобразование Галилея),}$$

$$U(x, t) = a u(a x, a^3 t) \text{ (растяжение).}$$

In[1]:=

```
Clear[u]
U = u[x + 6 a t, t] + a;
Expand[D[U, t] - D[U, x, x, x] - 6 U D[U, x]]

U = a^2 u[a x, a^3 t];
Factor[D[U, t] - D[U, x, x, x] - 6 U D[U, x]]
```

Out[3]= $u^{(0,1)}[6 a t + x, t] - 6 u[6 a t + x, t] u^{(1,0)}[6 a t + x, t] - u^{(3,0)}[6 a t + x, t]$

Out[5]= $a^5 (u^{(0,1)}[a x, a^3 t] - 6 u[a x, a^3 t] u^{(1,0)}[a x, a^3 t] - u^{(3,0)}[a x, a^3 t])$

Рациональное решение

Пусть все корни R совпадают:

$$(y')^2 = 4(y - a)^3, \quad c = 12 a.$$

Отсюда находим $y = a + (X + d)^{-2}$. Есть также особое решение $y = a$, но оно не представляет интереса.


Получаем полюсное решение

$$u(x, t) = a - 2(x - 12 a t + d)^{-2}.$$

```
In[6]:= u = -2 a - 2 (x - 12 a t + d)^-2;
Together[D[u, t] - D[u, x, x, x] - 6 u D[u, x]]

Plot[u /. {a -> 0, d -> 0}, {x, -10, 10}, PlotRange -> {{-10, 10}, {-10, 1}},
AxesLabel -> {"x", "u"}]
```

Out[7]= 0



Out[8]=

Солитон

Пусть совпадают два корня. С учётом преобразования Галилея, можно считать, что они равны 0, тогда имеем

$$(y')^2 = 4 y^2 (y - a), \quad c = 4 a.$$

Здесь удобно сделать замену $y = Y^{-2}$, это даст $(Y')^2 = 1 - a Y^2 \Rightarrow Y'' = -a Y$, что легко решается.

После несложных вычислений получаем три существенно различных решения, два для $a < 0$ и одно для $a > 0$:

```
In[9]:= y = - k^2 / Cosh[k X + d]^2;
Factor[TrigExpand[D[y, X]^2 - 4 y^2 (y + k^2)]]

y = k^2 / Sinh[k X + d]^2;
Factor[TrigExpand[D[y, X]^2 - 4 y^2 (y + k^2)]]

y = k^2 / Cos[k X + d]^2;
Factor[TrigExpand[D[y, X]^2 - 4 y^2 (y - k^2)]]

Clear[y]
```

Out[10]= 0

Out[12]= 0

Out[14]= 0


Возвращаясь к u , получаем солитон и два полюсных решения. С учетом растяжения и сдвига x , можно положить $k = 1, d = 0$ (но, конечно, когда мы перейдем к многосолитонным решениям, эти параметры у всех солитонов нельзя будет убить одновременно).

```
In[16]:= u =  $\frac{2 k^2}{\text{Cosh}[k x + 4 k^3 t + d]^2}$ ;

Together[TrigToExp[D[u, t] - D[u, x, x, x] - 6 u D[u, x]]]

Plot[u /. {k -> 1, d -> 0, t -> 0}, {x, -10, 10}, PlotRange -> {{-10, 10}, {-0.1, 2.1}},
  AxesLabel -> {"x", "u"}]
```

Out[17]= 0




```
In[19]:= u = -  $\frac{2 k^2}{\text{Sinh}[k x + 4 k^3 t + d]^2}$ ;

Together[TrigToExp[D[u, t] - D[u, x, x, x] - 6 u D[u, x]]]

Plot[u /. {k -> 1, d -> 0, t -> 0}, {x, -10, 10}, PlotRange -> {{-10, 10}, {-10, 1}},
  AxesLabel -> {"x", "u"}]
```


Out[20]= 0



```
In[22]:= u = -  $\frac{2 k^2}{\text{Cos}[k x - 4 k^3 t + d]^2}$ ;
Together[TrigToExp[D[u, t] - D[u, x, x, x] - 6 u D[u, x]]]

Plot[u /. {k -> 1, d -> 0, t -> 0}, {x, -10, 10}, PlotRange -> {{-10, 10}, {-5, 0.1}},
AxesLabel -> {"x", "u"}]
```

Out[23]= 0



Кноидальная волна

Рассмотрим кривую в фазовой плоскости (y, y')

$$(y')^2 = 4 R(y) = 4 y^3 - c y^2 - 2 c_1 y - c_2.$$

Её форма определяется нулями правой части. Если они вещественны и различны, то имеем овал и неограниченную ветвь. Если один корень вещественный и два комплексно сопряженных, то есть только одна неограниченная ветвь. На следующем графике можно посмотреть, как меняется кривая в зависимости от выбора коэффициентов (пунктиром нарисован график самого кубического многочлена $4 R$). После запуска программы нужно активировать эту ячейку [Shift-Enter].

In[25]=

```

p1[R_] := Plot[Evaluate[{4 R, 2 Sqrt[R], -2 Sqrt[R]} /. y -> yy], {yy, -4, 4},
  PlotStyle -> {{Dashed, Blue}, Red, Red},
  PlotRange -> {{-3, 3}, {-6, 6}},
  AxesLabel -> {"y", "y'"}]


```

```

Manipulate[
  p1[(4 y^3 - c y^2 - 2 c1 y - c2) / 4],
  {{c, 2}, -5, 5, Appearance -> "Labeled"},
  {{c1, 2}, -5, 5, Appearance -> "Labeled"},
  {{c2, -2}, -5, 5, Appearance -> "Labeled"}
]

```

Out[26]=



Двигая ползунки, можно получить и вырожденные случаи, рассмотренные выше. При этом, движению по неограниченным ветвям отвечают, в плоскости (x, u) , полюсные решения, а движению по петле на левом рисунке отвечает солитон.


In[27]=

```

GraphicsRow[{p1[y^2 (y + 2)], p1[y^2 (y - 2)]}, ImageSize -> 500]

```

Out[27]=



Рассмотрим теперь движение по ограниченному овалу кривой в случае, когда все корни вещественны и различны. Уравнение для y совпадает, с точностью до линейной замены, с уравнением для φ -функции

Вейерштрасса, но, так как эта функция имеет полюс в нуле, то удобнее выразить ответ через эллиптические функции Якоби. Напомним их определение: это решение системы ОДУ

$$\begin{aligned} \operatorname{sn}' x &= \operatorname{dn} x \operatorname{cn} x, \\ \operatorname{cn}' x &= -\operatorname{dn} x \operatorname{sn} x, \\ \operatorname{dn}' x &= -k \operatorname{sn} x \operatorname{cn} x, \\ \operatorname{sn} 0 &= 0, \operatorname{cn} 0 = \operatorname{dn} 0 = 1, \end{aligned}$$


где $0 < k < 1$. Из этих уравнений легко получить соотношения

$$\operatorname{sn}^2 x + \operatorname{cn}^2 x = 1, \quad k \operatorname{sn}^2 x + \operatorname{dn}^2 x = 1.$$

```
In[28]:=
cn = JacobiCN[x, k];
sn = JacobiSN[x, k];
dn = JacobiDN[x, k];
{D[sn, x] - dn cn, D[cn, x] + dn sn, D[dn, x] + k sn cn,
 D[sn^2 + cn^2, x], D[k sn^2 + dn^2, x]}

Plot[Evaluate[{sn, cn, dn} /. k -> 2], {x, 0, 10},
 PlotStyle -> {Red, Green, Blue, Black, Black, Black},
 PlotRange -> {-1.2, 1.2},
 PlotLegends -> LineLegend[{Red, Blue, Green}, {"sn x", "cn x", "dn x"}],
 AxesLabel -> {"x ", None}]
```

Out[31]= {0, 0, 0, 0, 0}



Нетрудно показать, что если положить $y = K_1 + K_2 \operatorname{sn}^2(K_3 x)$, то $(y')^2$ будет представляться в виде многочлена третьей степени от y . Остаётся только немного уточнить константы. Приведём сразу окончательный ответ:

```
In[33]:=
u = (4 b^2 (1 - 2 k) - c) / 6 + 2 b^2 k JacobiCN[b (x - c t) + d, k]^2;
Factor[D[u, t] - D[u, x, x, x] - 6 u D[u, x]]

Factor[% /. {JacobiDN[X_, k]^2 -> 1 - k JacobiSN[X, k]^2, JacobiCN[X_, k]^2 -> 1 - JacobiSN[X, k]^2}]
```

```
Out[34]= 16 b^5 k JacobiCN[d + b (-c t + x), k] JacobiDN[d + b (-c t + x), k]
JacobiSN[d + b (-c t + x), k] (1 - 2 k + 2 k JacobiCN[d + b (-c t + x), k]^2 -
JacobiDN[d + b (-c t + x), k]^2 + k JacobiSN[d + b (-c t + x), k]^2)
```

Out[35]= 0

Здесь можно выбрать постоянную c так, чтобы уничтожить свободный член, благодаря преобразованию Галилея, тогда график будет касаться оси x . За счет растяжения, можно положить

$b = 1$; параметр d уничтожается сдвигом x или t . Остаётся один существенный параметр k . При его увеличении пички в кноидальной волне разъезжаются друг от друга и в пределе $k = 1$ получается солитонное решение.

In[36]=

```
Manipulate[  
  Plot[2 k JacobiCN[x, k]^2, {x, -30, 30},  
    PlotStyle -> Blue, PlotRange -> {-0.1, 2}, AxesLabel -> {"x ", "u"}],  
  {{k, 0.2}, 0, 1, Appearance -> "Labeled"}  
]
```

Out[36]=

